

HT47C20 中 A/D 转换的使用

作者： 盛扬半导体（上海）有限公司软件部
日期： 2001/8/6
适用单片机： HT47C20、HT47C20L

本文主要介绍 HT47C20 单片机 A/D 转换器的应用及注意事项。

介绍

HT47C20 有两个 RC 型的 A/D 转换通道，包含两个可编程 16 位向上计数的计数器，计数器 A 的时钟来源可以是系统时钟、指令时钟（系统时钟/4）或 RTC 中断信号，计数器 B 的时钟来源是外部 RC 振荡电路。当 ADC/TM 位为“1”时（寄存器 ADRC 的第 1 位），TMRAL、TMRAH、TMRBL、TMRBH 组成了 A/D 转换器。

A/D 转换定时器 B 的时钟来源可以来自通道 0（IN0 外部时钟输入模式、RS0~CS0 振荡器、RT0~CS0 振荡器、CRT0~CS0 振荡器（CRT0 为一电阻）或 RS0~CRT0 振荡器（CRT0 为一电容））或来自通道 1（RS1~CS1 振荡器、RT1~CS1 振荡器或 IN1 外部时钟输入）。

与 A/D 转换有关的寄存器有 TMRAH、TMRAL、TMRC、TMRBH、TMRBL 和 ADRC。内部定时器时钟输入到 TMRAH 和 TMRAL 中，A/D 时钟输入到 TMRBH 和 TMRBL 中；赋初值时要先写入低字节，再写入高字节；读取时则要先读高字节，再读低字节。OV $\overline{\text{B}}$ /OV $\overline{\text{A}}$ 位（ADCR 寄存器的第 0 位）用来设置定时器 A 或定时器 B 溢出作为定时/计数器中断信号。在 A/D 转换模式下，当定时器 A（或定时器 B）溢出时 TON 位被清除并且计数器停止计数。

标志	位	功能
OV $\overline{\text{B}}$ /OV $\overline{\text{A}}$	0	在 RC 型 A/D 转换模式下，该位用来定义定时/计数器中断来自定时器 A 溢出或定时器 B 溢出（0=定时器 A 溢出，1=定时器 B 溢出） 在定时/计数器模式下，该位空缺
ADC/TM	1	设定定时/计数器或 RC 型 A/D 转换器允许（0=定时/计数器允许，1=A/D 转换器允许）
—	2~3	未定义，读取时为“0”
M0 M1 M2 M3	4 5 6 7	定义 A/D 转换器的工作模式（M3,M2,M1,M0） 0000=IN0 外部时钟输入模式 0001=RS0~CS0 振荡器（参考电阻和参考电容） 0010=RT0~CS0 振荡器（传感器电阻和参考电容） 0011=CRT0~CS0 振荡器（传感器电阻和参考电容） 0100=RS0~CRT0 振荡器（参考电阻和传感器电容） 0101=RS1~CS1 振荡器（参考电阻和参考电容） 0110=RT1~CS1 振荡器（传感器电阻和参考电容） 0111=IN1 外部时钟输入模式 1xxx=未定义

ADCR 寄存器

寄存器 ADCR 的 4~7 位用来决定选取哪一组电阻、电容来组成 TMRBH 和 TMRBL 的振荡输入。

寄存器 TMRC 的 TN0、TN1、TN2 用来决定定时器 A 的时钟来源。定时器 A 的时钟来源可以是系统时钟、指令时钟或实时时钟超时时钟。

当 TON 位（TMRC 的第 4 位）置为“1”时，定时器 A 和定时器 B 就开始计数，直到定时器 A（或定时器 B）发生溢出，此时，定时/计数器便置位中断请求标志（TF；INTC1 的第 4 位），同时计数器 A 和计数器 B 停止计数并 TON 位被清为“0”。


A/D 转换的使用

硬件部分：

传感器接至 CRT0 引脚，电容接至 CS0 引脚

*注意：用仿真器模拟时，用户板上的 Low_Volt 应接 VDD

电路图如下：



软件部分：

程序中设置 TIMER A 的时钟源为系统时钟，TIMER B 对外部 RC 振荡输入计数，并选择 TIMER A 溢出作为中断。

程序清单：

```
include ht47c20.inc
data .section 'data'
count1 db ?          ; count1、count2 用来保存 TIMER B 的计数值
count2 db ?

code .section at 0 'code'
    org    00h
    jmp    start
    org    04h
```

```

    reti
    org    08h
    reti
    org    0ch
    reti
    org    10h
    jmp    ad_int          ; 定时/计数器中断入口
; -----
start:
    clr    intc0
    clr    intc1
    mov    a, 0fh          ; 设置 TIMER A 计数模式
    andm   a, tmrc          ; TIMER A 的时钟来源为系统时钟
    mov    a, 32h          ; A/D 转换允许
    mov    adcr, a          ; 并设置 TIMER B 时钟来源为 CRT0-CS0 振荡输入
    clr    acc              ; 定时/计数器赋初值
    mov    tmral, a
    mov    tmrah, a
    mov    tmrbl, a
    mov    tmrbh, a
    set    tmrc.4           ; 打开计数器, TIMER A 和 TIMER B 开始计数
    set    intc1.0          ; 定时/计数器中断允许
    set    intc0.0          ; 总中断允许
    jmp    $
; -----
ad_int:                      ; 定时/计数器中断服务子程序
    mov    a, tmrbh          ; 读取 TIMER B 高字节内容
    mov    count1, a
    mov    a, tmrbl          ; 读取 TIMER B 低字节内容
    mov    count2, a
    call   calculate         ; 计算
    reti
; -----
calculate proc
    ; 计算过程
    ret
calculate endp

```

下面介绍一种 A/D 转换的计算方法。

以第一通道为例, 当振荡源为参考电阻 R_S 、参考电容 C_S 时, 设频率为 f_{RSCS} , 有:

$$f_{RSCS} = N / (R_S \times C_S) \text{ ----- (1)}$$

N ——常数

同样, 当振荡源为传感器电阻 R_T 、参考电容 C_S 时, 有:

$$f_{RTCS} = N / (R_T \times C_S) \text{ ----- (2)}$$

设计数时间为 T，则在 T 时间内，检测到的振荡次数为

$$K=f \times T \text{-----} (3)$$

由 (1)、(3) 式得：

$$K_{RSCS}=N \times T / (RS \times CS) \text{-----} (4)$$

由 (2)、(3) 式得：

$$K_{RTCS}=N \times T / (RT \times CS) \text{-----} (5)$$

(4)、(5) 式两边分别相比得：

$$RT=K_{RSCS} \times RS / K_{RTCS} \text{-----} (6)$$

这就是传感器电阻的计算公式，由此公式可见，传感器 RT 的测量值与参考电容无关。所以 RT 的测量值与参考电容的精度无关，与参考电阻 RS 的精度有关。由时间 T 内计到的 RS~CS 的振荡次数 K_{RSCS} ，时间 T 内计到的 RT~CS 的振荡次数 K_{RTCS} ，即可算出 RT 的值。

如果要测量传感器电容时，同样可得到：

$$CT=K_{RSCS} \times CS / K_{RSCS} \text{-----} (7)$$

注意，参考电容应为精密电容。

(6) 式虽然可得到 RT 的值，但是要算一个乘法、一个除法，都是比较复杂的运算，在 (6) 式中，要是能让

$$RS / K_{RTCS}=2^n \text{-----} (8)$$

则得：

$$RT=K_{RSCS} \times 2^n \text{-----} (9)$$

下面，我们根据 HT47C20 的特点，介绍一种方式，使得 RT 可以根据 (9) 式算出。

1、预估 RT 的值决定 n：

以 RT、CS 为振荡源，测量在时间 T_x 内 IN0 口输入的脉冲个数 K_{RT0CS} 。根据 K_{RT0CS} 来预估 RT 的值从而决定 n 的值。

这里，n 的值主要是由用户根据所需传感器的测量速度决定的，我们建议 $n=0\sim 8$ 。或者 $n=-1\sim 7$ 。用户可根据自己的需要选择。这样我们可以将 T_x 时间内测量到的脉冲次数分成 9 档，存入表格，测量出 K_{RT0CS} 的值时，查表就可得出对应的 n。 T_x 可选择在 0.5 秒之内。用户可根据需要自己定。一般只要能决定 n 的值， T_x 越小越好。

2、决定需要测量的脉冲 K_{RTCS} ：

n 值定好后，由 (8) 式得：

$$K_{RTCS}=RS/2^n \text{-----} (10)$$

四舍五入取 $RS/2^n$ 的整数为 K_{RTCS} 。

现在回到 (6) 式子，RS 为已知， K_{RTCS} 已经找到，并且 $RS / K_{RTCS}=2^n$ ，现在，只要找到 K_{RSCS} ，就能由 (9) 式得到 RT。

这里需要注意的是，由 (3)、(4) 式可以看出，测量 K_{RTCS} 和 K_{RSCS} 的时间必须一样。假设测量的时间为 T_0 。

3、找出 T_0 ：

以 TIMER B 计数，TIMER A 定时，以 RT、CS 为振荡源。TIMER B 初始值为 $(65536 - K_{RTCS})$ ，TIMER A 初始值为 0h，并设 TIMER B 溢出作为中断。发生中断后，读取 TIMER A 中的值，即为 T_0 。做这些工作时，要注意定时/计数器读/写的先后顺序。

4、寻找 K_{RSCS} ：

现在， T_0 已经定好了，我们就可以找出在 T_0 时间内，RS、CS 的振荡次数 K_{RSCS} 。

以 TIMER B 计数，TIMER A 定时，以 RS、CS 为振荡源。TIMER A 初始值为 $(65536 - T_0)$ ，TIMER B 初始值为 0h，并设 TIMER A 溢出作为中断。发生中断后，读取 TIMER B 中的值，即为 K_{RSCS} 。

5、得出 RT：

回到（9）式，所有的值都已经得到，只要将 K_{RSCS} 左移或右移 n 位，就可以得到 R_T 的值。 R_T 的值得出后还没有大功告成。根据我们实验室得出的数据，即使传感器没有误差，用 HT47C20 测量出来的值还是和实际的值有一定的误差。结果如下图：




图 1：RT 测量值和实际值对照表

由表可看出， R_T 在接近参考电阻 R_S 的地方，误差最小，在远离参考电阻 R_S 的地方误差最大。我们可以根据测量到的电阻值对照上图进行校正。

6、校正：

如图，我们将曲线分成线性的 3 段： R_T 从 1000~10k、从 10k~1M、从 1M~10M。测出 R_T 后，找出在三段曲线中的哪一部分，用内差法校正。这样校正要计算，会比较麻烦。另一种方法查表法可能会相对简单。

用传感器测量温度时，一般会在传感器电阻值和对应温度之间建一个表格，在建表格时，可以先考虑到 HT47C20 测量误差，这样测出 R_T 直接去查表就可以直接消除误差。

例程：

```
include ht47c20.inc
data .section 'data'
Rs      equ    0064h          ; 参考电阻 100KΩ
n        db    ?
count   db    ?
kh       db    ?              ; 计数值 K
kl       db    ?
th       db    ?              ; 计数时间 T
tl       db    ?
int_count db    ?             ; 中断次数计数器
; -----
code .section at 0 'code'
```

```

    org    00h
    jmp    start
    org    04h
    reti
    org    08h
    reti
    org    0ch
    reti
    org    10h                ; 定时/计数器中断入口
    jmp    t_int

; -----
start: clr    intc0
      clr    intc1
      clr    adcr
      set    adcr.1          ; A/D 转换允许
      mov    a, 0fh          ; TIMER A 计数系统时钟
      andm   a, tmrc
      set    adcr.4          ; TIMER B 计数传感器电阻-参考电容
      set    adcr.5
      clr    acc             ; 定时/计数器初值
      mov    tmral, a
      mov    tmrah, a
      mov    tmrbl, a
      mov    tmrbh, a
      clr    int_count       ; 中断次数计数器清零
      set    tmrc.4          ; 开定时/计数器
      set    intc1.0         ; 定时/计数器中断允许
      set    intc0.0         ; 总中断允许
      jmp    $

; -----
t_int: inc    int_count       ; 定时/计数器中断服务子程序
      mov    a, int_count    ; 中断次数判断
      mov    count, a
      sdz    count
      jmp    next
      jmp    int1
next:  sdz    count
      jmp    int3
      jmp    int2

; -----
int1:  mov    a, tmrbh        ; 第一次中断
      mov    kh, a           ; 读取 TIMER B 的值, 记为 KRT0CS
      mov    a, tmrbl
      mov    kl, a

```

```

        clr    n
        clr    tblp
scan_tab:                                ; 查表, 确定 n 的值
        tabrdl acc
        sub    a, kl
        mov    a, tblh
        sbc    a, kh
        inc    n
        inc    tblp
        sz     c
        jmp    scan_over
        jmp    scan_tab
scan_over:                                ; 结束查表
        dec    n
        mov    a, high Rs
        mov    kh, a
        mov    a, low Rs
        mov    kl, a
        mov    a, n
        mov    count, a
        sz     count                    ; 判断 n 是否为 0
        jmp    div                     ; n 不为 0, 则进行除法操作
        jmp    div_over                ; n 为 0, 则不进行除法操作
div:     clr    c                       ; 除法操作
        rrc    kh
        rrc    kl
        sdz    count
        jmp    div
div_over:                                ; 结束除法操作
                                           ; 第二次计数初始化
        clr    acc                    ; TIMER A 计数初值为 0
        mov    tmral, a
        mov    tmrah, a
        mov    a, 0ffh                ; TIMER B 计数初值为 65536-KRTCS
        sub    a, kl
        mov    tmrbl, a
        mov    a, 0ffh
        sub    a, kh
        mov    tmrbh, a
        set    adcr.0                  ; TIMER B 溢出作为中断
        set    tmrc.4
        reti
; -----
int2:    mov    a, tmrah                ; 第二次中断

```

```

        mov     th, a                ; 读取 TIMER A 的值, 记为 T0
        mov     a, tmral
        mov     tl, a

                                ; 第三次计数初始化
        clr     adcr.0               ; TIMER A 溢出作为中断
        clr     adcr.5               ; TIMER B 计数参考电阻-参考电容
        mov     a, 0ffh              ; TIMER A 计数初值为 65536-T0
        sub     a, tl
        mov     tmral, a
        mov     a, 0ffh
        sub     a, th
        mov     tmrah, a
        clr     acc                   ; TIMER B 计数初值为 0
        mov     tmrbl, a
        mov     tmrbh, a
        set     tmrc.4
        reti

; -----
int3:   mov     a, tmrbh              ; 第三次中断
        mov     kh, a                ; 读取 TIMER B 的值, 记为 KRSCS
        mov     a, tmrbl
        mov     kl, a
        mov     a, n
        mov     count, a
        sz      count                ; 判断 n 是否为 0
        jmp     mul                  ; n 不为 0, 则进行 乘法操作
        jmp     mul_over              ; n 为 0, 则不进行乘法操作
mul:    clr     acc                   ; 乘法操作
        rlc     kl
        rlc     kh
        sdz     count
        jmp     mul
mul_over:
        call    display              ; 调用显示子程序
                                ; 结束一次测量, 为下一次测量做初始化
        clr     adcr.0               ; TIMER A 溢出作为中断
        set     adcr.5               ; TIMER B 计数传感器电阻-参考电容
        clr     acc                   ; 定时/计数器初值
        mov     tmral, a
        mov     tmrah, a
        mov     tmrbl, a
        mov     tmrbh, a
        clr     int_count             ; 清除中断次数计数器
        set     tmrc.4

```



```
        reti
; -----
display proc                ; 显示子程序
        ret
display endp
; -----
        org    700h
tab:    dc      100h, 200h, 400h, 800h, 1000h, 2000h, 4000h, 8000h, 0ffffh
```